Getting Started in Plotting in 3D

Note: You may notice differences between this Maple worksheet and the equivalent Mathematica notebook. These differences were introduced to preserve the content of these modules and were necessary because of major functional differences between Maple and Mathematica.

Introduction

OBJECTIVE: To learn to use *Maple* to plot lines, cylinders, and quadric surfaces.

In this module, you will learn how *Maple* can help you plot the lines, planes, cylinders, and quadric surfaces you have been reading about. In the process of plotting lines and planes, you will gain insight into their vector definitions.

Technology Guidelines

NOTE: If you have just finished a worksheet, **restart** *Maple* before executing a new worksheet.
TO OPEN SECTIONS,
Click on the **PLUS** sign at the left hand side of the screen *or* select **Expand All Sections** from the **View** drop down menu.

TO STOP AN EXECUTION
Click on **STOP** button from the toolbar.

ORDER OF EXECUTION
Execute commands in the order given. Do not skip any *Maple* Input lines within a given worksheet

Alternatively, you can execute the entire worksheet by selecting the **Execute Worksheet** command from the **Edit** drop down menu.

SAVING WORKSHEETS.

You can save anytime to any directory you choose, and it is wise to save often.
EXPERIENCING MAJOR PROBLEMS
Save if appropriate and then shut down *Maple* and start it up again.

Part I: Lines and Planes

▪ Interpreting Lines Using Their Vector Definition

You can construct straight lines in two and three dimensions using the vector definition. Given two points on the line, we will first determine the direction of the line and then write any other point on the line as the position vector to that point plus a multiple (*t*) of the vector in the direction of the line. We will start with two points, finding the direction determined by those two points, and then writing and plotting the parametric equations for the line in 3-space.

Click anywhere on the graph to rotate the plot on all three axes.

```
> restart:
```

```
> o:=[0,0,0]:
  p1:=[1,5,-6]:
  p2:=[4,2,5]:
  dir:=p1-p2:
  print(`The direction of the line is`, dir);
  print(`[x,y,z]=`, expand((p1-o)+t*dir));
  plot3d([(p1[1]-o[1])+t*dir[1],(p1[2]-o[2])+t*dir[2], (p1[3]-o[3])+t*dir[3]], t=-10..10, s=
  axes=FRAME, labels=[x,y,z]);
```

The direction of the line is, [-3, 3, -11]

[x,y,z]=, $[-3\,t+1,\ 3\,t+5,\ -11\,t-6]$



▪ Constructing Planes Using Vectors

We begin by constructing a plane from three points on the plane. We use those three points to determine two directions in the plane and then compute other points on the plane by adding their position vector to a linear combination (using parameters *s* and *t*) of the known vectors in the plane.

```
> o:=[0,0,0]:
  p1:=[2,3,-1]:
  p2:=[8,-2,5]:
  p3:=[-4,0,9]:
  pts:=expand((p1-o)+s*(p2-p1)+t*(p3-p2)):
  print(`[x,y,z]=`,pts );
```

$$[x,y,z]=, [-12\,t+6\,s+2, 2\,t-5\,s+3, 4\,t+6\,s-1]$$

We can also find the equation of the plane using the cross product to find the normal vector to the plane. Then we can find the points (*x, y, z*) that lie in the plane by dotting any vector in the plane into that normal vector and setting the dot product equal to 0.

```
> vec1:=p2-p1:
  vec2:=p3-p2:
  norma:=linalg[crossprod](vec1,vec2):
  eq:=linalg[dotprod](norma, ([x,y,z]-p1))=0:
  eqz:=solve(eq, z);
  plot3d(eqz, x=-10..10, y=-10..10, labels=["x","y","z"], axes=BOXED);
```

$$eqz := \frac{19}{3}-\frac{2\,x}{3}-2\,y$$



Note that the relationships among *x, y,* and *z* are exactly the same in both parametric and nonparametric form.

```
> x1:=pts[1];
  y1:=pts[2];
  z1:=pts[3];

  if 19/3-2*y1-2/3*x1=z1 then print(`True`) else print(`False`) fi;;
```

$$x1 := -12\,t+6\,s+2$$

$$y1 := 2\,t-5\,s+3$$

$$z1 := 4\,t + 6\,s - 1$$

True

You Try It: Part I

▪ Finding equations of lines

To find the equations of lines and planes, just change the numbers in the points given, then re-execute the commands to see other lines. You can replace the red items in the following input commands to create a new line.

```
> o:=[0,0,0]:
  p1:=[2,-3,0]:
  p2:=[-1,8,4]:
  dir:=p1-p2:
  eqn:=expand((p1-o)+t*dir):
  print(`The direction of the line is`, dir);
  print(`[x,y,z]=`,eqn);
  plot3d([(p1[1]-o[1])+t*dir[1],(p1[2]-o[2])+t*dir[2], (p1[3]-o[3])+t*dir[3]], t=-10..10, s=
  axes=FRAME, labels=[x,y,z]);
```

The direction of the line is, [3, -11, -4]

[x,y,z]=, $[3\,t + 2, -11\,t - 3, -4\,t]$



▪ Finding equations of planes

Now, pick any three points, replace the red numbers with your points, and execute the cell to see the plane determined by the three points.

```
> o:=[0,0,0]:
  p1:=[2,5,1]:
  p2:=[0,3,-4]:
  p3:=[1,0,7]:
  pts:=expand((p1-o)+s*(p2-p1)+t*(p3-p2)):
  print(`[x,y,z]=`,pts);
  vec1:=p2-p1:
  vec2:=p3-p2:
  norma:=linalg[crossprod](vec1,vec2):
  eq:= linalg[dotprod](norma, ([x,y,z]-p1))=0:
  solz:=solve(eq,z);
  plot3d(solz(x,y,z), x=-10..10, y=-10..10, axes=BOXED, labels=[x,y,z]);
```

$$[x,y,z]=, [t-2s+2, -3t-2s+5, 11t-5s+1]$$

$$solz := \frac{19}{8}+\frac{37x}{8}-\frac{17y}{8}$$



Part II: Cylinders and Quadric Surfaces

When you look at the equations for cylinders or quadric surfaces, notice that they are not in the standard form for graphing on most graphing calculators or software packages, because you cannot usually solve explicity for the z variable. To get around this, we use a special command called **implicitplot3d** after we first read in a package to enable it to work. In the next chapter (Chapter 11), you will learn about contour plots, but, for now, just think of this as a command to get the job done.

Suppose that you want to plot the cylinder $y = x^2$. We think of this as $y - x^2 = 0$, then we ask for the **implicitplot3d** of $y - x^2$. The default is to plot only the contour when the function specified is 0. Note that we must specify the values of *x, y,* and *z* over which to extend our plot.

```
> with(plots):
```

Warning, the name changecoords has been redefined

```
> f1:=y-x^2;
  implicitplot3d(f1, x=-3..3, y=-2..2, z=-4..4, axes=FRAME, labels=["x","y","z"]);
```

$$f1 := y - x^2$$



Let's try another cylinder.

```
> f2:=y^2-4*z^2-1:
  implicitplot3d(f2,x=-2..2, y=-2..2, z=-2..2, axes=FRAME, labels=["x","y","z"]);
```



Next, we will plot the quadric surface, $z^2 - 2x^2 - 3y^2 = 0$.

```
> f3:=z^2-2*x^2-3*y^2=0;
  implicitplot3d(f3, x=-3..3, y=-2..2, z=-4..4, axes=FRAME, labels=["x","y","z"], grid=
  [15,15,15]);
```

$$f3 := z^2 - 2x^2 - 3y^2 = 0$$



Here's an ellipsoid.

```
> f4:=x^2+4*y^2+z^2-4;
  implicitplot3d(f4, x=-2..2, y=-1..1, z=-2..2, labels=["x","y","z"], axes=FRAME);
```

$$f4 := x^2 + 4\,y^2 + z^2 - 4$$



You Try It: Part II

To plot the cylinders and quadric surfaces, go back to any of the plots and alter the functions as you wish. Be sure to use correct terminology. Note that some 3D plots take quite a bit of time to plot. If you find you are waiting too long, you can always go to the toolbar and click on STOP to abort the operation.. To experiment, replace the function, f5, with another function of *x, y*, and *z*. Be certain to use correct terminology. You might want to see what this one looks like first, before putting in a different function. Why isn't this a quadric surface?

```
> f5:=z^3-2*x^2+3*y^2;
  plots[implicitplot3d](f5, x=-3..3, y=-2..2, z=-4..4, labels=["x","y","z"], axes=FRAME, grid=
  [15,15,15]);
```

$$f5 := z^3 - 2\,x^2 + 3\,y^2$$


4
2
z 0
-2
-4
-2
-1
0
y
1
2
3
2
1
0
x
-1
-2
-3

> ?

>